' Record Class Info - © 1993-2002 Aiman Hanna
' This program explores Files. The program shows how files can be created, opened, written and closed. 

' Important Note: When you open a file for output (to write to), you must make sure that the file does not

' exist before you create it. Creating an existing file will simply destroy the file (removes all its contents). 

' This problem will occur even if you use "ForAppending" when you open the stream; see below for details. 

' Key Points:

'       1) Files

'       2) FileSystemObject, File & TextStream types

'       3) FileExists, CreateTextFile, GetFile methods (functions) 

'            of the FileSystemObject type (actually class)

'       4) OpenAsTextStream function of the File type

'       5) Close function of the TextStream type

'       6) Form_Load() and Form_Terminate() subs
Option Explicit

Dim fso As New FileSystemObject

Dim f1 As File

Dim txtstrm As TextStream

Private Sub Form_Load()

   ' This sub will be called automatically (and only once)

   'when the form is loaded at run-time (when the program starts)

   ' Create a text file called c201_info.dat under c:\comp201 directory

   ' However, create the file only if it does not exist

   If fso.FileExists("c:\comp201\c201_info.dat") = False Then

       Call fso.CreateTextFile("c:\comp201\c201_info.dat")

   End If

   ' Once the file is created, reference the file

   Set f1 = fso.GetFile("c:\comp201\c201_info.dat")

   ' Open a text stream for writing to the file

   Set txtstrm = f1.OpenAsTextStream(ForWriting)   ' Use "ForAppending" if need to keep the old data and only add to it

   ' Display path in lblFileName

   lblFileName.Caption = "All information will be recorded in: " & f1.Path

End Sub

Private Sub cmdRecord_Click()

   ' Write the student info to the file

   Call txtstrm.WriteLine(txtID.Text & vbTab & _

                             txtFN.Text & vbTab & _

                             txtGN.Text & vbTab & _

                             txtMarks.Text)

   ' Clear the controls

   txtID.Text = ""

   txtFN.Text = " "

   txtGN.Text = " "

   txtMarks.Text = " "

End Sub

Private Sub Form_Terminate()

' This sub will be called automatically when the program terminates

    ' Before exiting close the file

    Call txtstrm.Close

End Sub

Private Sub cmdExit_Click()

    End

End Sub

